
1 Introduction
[bookmark: OLE_LINK91][bookmark: OLE_LINK90][bookmark: OLE_LINK6]In RAN4 #90bis, how to reduce TBDs derived according to the agreed rule for NR performance requirements are intensively discussed and an update was agreed by extending the threshold of impairment result span from 3dB to 4dB [1]. And also it was discovered the previous excel script [2] continues the iteration when the selected impairment result span is larger than the threshold, which is not exactly what the agreed procedure assumes. Therefore we update the excel script accordingly.
2 Discussion
The script is shown below:
--------------------------------------

Function DerivingPerfReq(SimVal_in As Range) As Double
    Dim i, N, M As Long
    Dim Ideal(50) As Double
    Dim Impairment(50) As Double
    Dim Min_ideal, Max_ideal, Min_impairment, Max_impairment As Double
    Dim Min_ideal_pos, Max_ideal_pos, Min_impairment_pos, Max_impairment_pos As Integer
    Dim Avg_ideal, Avg_impairment As Double
    Dim SimVal(100) As Double
    Dim Avg As Double
    Dim Avg_pos As Double
    Dim Company(50) As Integer
    Dim row_index, col_index As Integer
    
    
    
    ' Validating inputs
    N = SimVal_in.Count
    If (Int(N / 2) * 2) <> N Then
        'MsgBox ("Simulation results must be provided in pairs of ideal and impairment results")
        DerivingPerfReq = -101
        Exit Function
    End If
        
    
    row_index = SimVal_in.Row
    col_index = SimVal_in.Column
    
    M = 0 ' Number of effective pairs
    For i = 0 To N / 2 - 1
        If IsEmpty(SimVal_in.Value2(1, 2 * i + 1)) Then
            
        ElseIf IsEmpty(SimVal_in.Value2(1, 2 * i + 2)) Then
        ElseIf IsNumeric(SimVal_in.Value2(1, 2 * i + 1)) And IsNumeric(SimVal_in.Value2(1, 2 * i + 2)) Then
            SimVal(2 * M) = SimVal_in(2 * i + 1)
            SimVal(2 * M + 1) = SimVal_in(2 * i + 2)
            M = M + 1
        End If
    Next i
        
    If M < 3 Then
        'MsgBox ("Not enough Simulations: at least 3 companies")
        DerivingPerfReq = -100
        Exit Function
    End If
                
                
    Do While M >= 3
        Min_ideal = SimVal(0)
        Max_ideal = SimVal(0)
        Min_impairment = SimVal(1)
        Max_impairment = SimVal(1)
        
        Min_ideal_pos = 0
        Max_ideal_pos = 0
        Min_impairment_pos = 1
        Min_impairment_pos = 1
        
        Avg_ideal = SimVal(0)
        Avg_impairment = SimVal(1)
    
        ' Find the spans and average for each
        For i = 1 To M - 1
             If Min_ideal > SimVal(2 * i) Then
                Min_ideal = SimVal(2 * i)
                Min_ideal_pos = i
             End If
             If Max_ideal < SimVal(2 * i) Then
                Max_ideal = SimVal(2 * i)
                Max_ideal_pos = i
             End If
             If Min_impairment > SimVal(2 * i + 1) Then
                Min_impairment = SimVal(2 * i + 1)
                Min_impairment_pos = i
             End If
             
             If Max_impairment < SimVal(2 * i + 1) Then
                Max_impairment = SimVal(2 * i + 1)
                Max_impairment_pos = i
             End If
             
             Avg_ideal = Avg_ideal + SimVal(2 * i)
             Avg_impairment = Avg_impairment + SimVal(2 * i + 1)
             
        Next i
        Avg_ideal = Avg_ideal / M
        Avg_impairment = Avg_impairment / M
        
        ' Check if ideal span is within 2 dB
        If (Max_ideal - Min_ideal) <= 2 Then
            Exit Do
        Else: ' remove outlier
                ' 1) find outlier
                Avg = Abs(SimVal(0) - Avg_ideal)
                Avg_pos = 0
                For i = 1 To M - 1
                    If Abs(SimVal(2 * i) - Avg_ideal) > Avg Then
                        Avg = Abs(SimVal(2 * i) - Avg_ideal)
                        Avg_pos = i
                    End If
                Next i
                ' 2) excluding the outlier
                For i = Avg_pos To M - 2
                    SimVal(2 * i) = SimVal(2 * i + 2)
                    SimVal(2 * i + 1) = SimVal(2 * i + 3)
                Next i
                ' M dec
                M = M - 1
        End If
        
    Loop ' iterated until ideal result span is within the threshold
        
    ' either M < 3 or ideal results span met
    If M < 3 Then
            DerivingPerfReq = -103 ' Not enough input after outlier(s) removal
    'check span for impairment results, threshold changed from 3dB to 4dB
    ElseIf (Max_impairment - Min_impairment) <= 4 Then
            DerivingPerfReq = Round(Avg_impairment, 1)
    Else:
            DerivingPerfReq = -102 ' Too large span for impairment results
    End If
            
End Function

